Full version of a paper presented at *The 2nd World Congress on Logic and Religion*, Warsaw, Poland, June 2017

A Mechanically Assisted Examination of Begging the Question in Anselm's Ontological Argument

John Rushby
Computer Science Laboratory
SRI International, Menlo Park CA USA

Abstract

I use mechanized verification to examine several first- and higher-order formalizations of Anselm's Ontological Argument against the charge of begging the question. I propose three different criteria for a premise to beg the question in fully formal proofs and find that one or another applies to all the formalizations examined. My purpose is to demonstrate that mechanized verification provides an effective and reliable technique to perform these analyses; readers may decide whether the forms of question begging so identified affect their interest in the Argument or its various formalizations.

1 Introduction

I assume readers have some familiarity with St. Anselm's 11'th Century Ontological Argument for the existence of God; a simplified translation from the original Latin of Anselm's *Proslogion* is given in Figure 1, with some alternative readings in square parentheses. This version of the argument appears in Chapter II of the Proslogion; another version appears in Chapter III and speaks of the *necessary* existence of God. Many authors have examined the Argument, in both its forms; in recent years, most begin by rendering it in modern logic, employing varying degrees of formality. The Proslogion II argument is traditionally rendered in first-order logic while propositional modal logic is used for that of Proslogion III. More recently, higher-order logic and quantified modal logic have been applied to the argument of Proslogion II. My focus here is the Proslogion II argument, represented completely formally in first- or higher-order logic, and explored with the aid of a mechanized verification system. Elsewhere [15], I use a verification system to examine renditions of the argument in modal logic, and also the argument of Proslogion III.

Verification systems are tools from computer science that are generally used for exploration and verification of software or hardware designs and algorithms; they comprise a specification language, which is essentially a rich (usually higher-order)

We can conceive of [something/that] than which there is no greater

If that thing does not exist in reality, then we can conceive of a greater thing—namely, something [just like it] that does exist in reality

Thus, either the greatest thing exists in reality or it is not the greatest thing

Therefore the greatest thing exists in reality

[That's God]

Figure 1: The Ontological Argument

logic, and a collection of powerful deductive engines (e.g., satisfiability solvers for combinations of theories, model checkers, and automated and interactive theorem provers). I have previously explored renditions of the Argument due to Oppenheimer and Zalta [11] and to Eder and Ramharter [8] using the PVS verification system [13, 14], and that provides the basis for the work reported here. Benzmüller and Woltzenlogel-Paleo have likewise explored modal arguments due to Gödel and Scott using the Isabelle and Coq verification systems [4, 5].

Mechanized analysis confirms the conclusions of most earlier commentators: the Argument is valid. Attention therefore focuses on the premises and their interpretation. The premises are *a priori* (i.e., armchair speculation) and thus not suitable for empirical confirmation or refutation: it is up to the individual reader to accept or deny them. We may note, however, that the premises are consistent (i.e., they have a model), and this is among the topics that I previously subjected to mechanized examination [13] (as a byproduct, this examination demonstrates that the Argument does not compel a theological interpretation: in the exhibited model, that "than which there is no greater" is the number zero).

The Argument has been a topic of enduring fascination for nearly a thousand years; this is surely due to its derivation of a bold conclusion from unexceptionable premises, which naturally engenders a sense of disquiet: "The Argument does not, to a modern mind, seem very convincing, but it is easier to feel that it must be fallacious than it is to find out precisely where the fallacy lies" [16, p. 472]. Many commentators have sought to identify a fallacy in the Argument or its interpretation (e.g., Kant famously denied it on the basis that "existence is not a predicate"). One direction of attack is to claim that the Argument "begs the question"; that is, it essentially assumes what it sets out to prove [12, 19].[1] This is the charge that I examine here.

[1]The phrase is widely misunderstood to mean "to invite the question"; use of the phrase in logic derives from medieval translations of Aristotle.

Begging the question has traditionally been discussed in the context of informal or semi-formal argumentation and dialectics [2, 17, 20, 21], where the concern is whether arguments that beg the question should be considered fallacious, or merely unpersuasive, or could even be persuasive. Here, we examine question begging in the context of fully formal, mechanically checked proofs. My purpose is to provide techniques that can identify potential question begging in a systematic and fairly unequivocal manner. I do not condemn the forms of question begging that are identified; rather, my goal is to highlight them so that readers can make up their own minds and can also use these techniques to find other cases.

The paper is structured as follows. In the next section, I introduce a strict definition of "begging the question" and show that a rendition of the Argument due to Oppenheimer and Zalta [11] is vulnerable to this charge. Oppenheimer and Zalta use a definite description (i.e., they speak of "*that* than which there is no greater") and require an additional assumption to ensure this is well-defined. Eder and Ramharter argue that Anselm did not intend this interpretation (i.e., requires only "*something* than which there is no greater") [8, Section 2.3] and therefore dispense with the additional assumption of Oppenheimer and Zalta. In Section 3, I show that this version of the argument does not beg the question under the strict definition, but that it does under a plausible weakening. In Section 4, I consider an alternative premise due to Eder and Ramharter and show that this does not beg the question under either of the previous interpretations, but I argue that it is at least as questionable as the premise that it replaces because it so perfectly discharges the main step of the proof that it seems reverse-engineered. I suggest a third interpretation for "begging the question" that matches this case. In Section 5, I consider the higher-order treatment of Eder and Ramharter [8, Section 3.3] and a variant derived from Campbell [6]; these proofs are more complicated than the first-order treatments but I show how the third interpretation for "begging the question" applies to them. Finally, in Section 6, I discuss some possible implications of this work.

2 Begging the Question: Strict Case

"Begging the question" is a form of circular reasoning in which we assume what we wish to prove. It is generally discussed in the context of informal argumentation where the premises and conclusion are expressed in natural language. In such cases, the question-begging premise may state the same idea as the conclusion, but in different terms, or it may contain superfluous or even false information, and there is much literature on how to diagnose and interpret such cases [2,3,17,20–22]. That is not my focus. I am interested in formal, deductive arguments, and in criteria for begging the question that are themselves formal. Now, deductive proofs do not generate new knowledge—the conclusion is always implicit in the premises—but they can generate surprise or insight; I propose that criteria for question begging should focus on the extent to which either the conclusion or its proof are "so directly" represented in the premises as to vitiate the hope of surprise or insight.

The basic criterion for begging the question is that one of, or a collection of, the premises is equivalent to the conclusion. But if some of the premises are equivalent to the conclusion, what are the other premises for? Certainly we must need all the premises to deduce the conclusion (else we can eliminate some of them); thus we surely need all the other premises before we can establish that some of them are equivalent to the conclusion. Hence, the criteria for begging the question should apply *after* we have accepted the other premises. Thus, if C is our conclusion, Q our "questionable" premise (which may be a conjunction of simpler premises) and P our other premises, then Q begs the question if C is equivalent to Q, assuming P: i.e., $P \vdash C = Q$. Of course, this means we can prove C using Q: $P, Q \vdash C$, and we can also do the reverse: $P, C \vdash Q$.

```
oandz: THEORY
BEGIN

  beings: TYPE
  x, y: VAR beings

  >: (trichotomous?[beings])

  God?(x): bool = NOT EXISTS y: y > x

  re?(x): bool

  ExUnd: AXIOM EXISTS x: God?(x)

  Greater1: AXIOM FORALL x: (NOT re?(x) => EXISTS y: y > x)

  God_re: THEOREM re?(the(God?))

%---------------- Question Begging Analysis ----------------------

  Greater1_circ: THEOREM FORALL x: (NOT re?(x) => EXISTS y: y > x)

  G1_begs_question: THEOREM
        re?(the(God?)) = FORALL x: (NOT re?(x) => EXISTS y: y > x)

END oandz
```

Figure 2: Oppenheimer and Zalta's Treatment, in PVS

Figure 2 presents Oppenheimer and Zalta's treatment of the Ontological Argument [11] in PVS. I will not describe this in detail, since it is explained at tutorial level [13] and used [14] elsewhere, but remark that the identifiers and constructs used here are from [14] rather than [13]. Briefly, the specification language of PVS is a strongly typed higher-order logic with predicate subtypes, although this example uses only first order. The uninterpreted type `beings` is used for those things that are "in the understanding" (i.e., "understandable beings"). Note that a question mark at the end of an identifier is merely a convention to indicate predicates (which in PVS are simply functions with return type `bool`). The predicate `God?` recognizes those beings "than which there is no greater"; the axiom `ExUnd` asserts the existence of at least one such being; `the(God?)` is a definite description that identifies this being. PVS generates a proof obligation (not shown here) to ensure this being is unique (this is required by the predicate subtype used in the definition of `the`, which is part of the "Prelude" of standard theories built in to PVS) and `ExUnd` and the trichotomy (also from the Prelude) of `>` are used to discharge this obligation.[2] The uninterpreted predicate `re?` identifies those beings that exist "in reality" and the axiom `Greater1` asserts that if a being does not exist in reality, then there is a greater being.

The theorem `God_re` asserts that the being identified by the definite description `the(God?)` exists in reality. The PVS proof of this theorem is accomplished by the following commands.

PVS Proof
```
(typepred "the(God?)")  (use "Greater1") (grind)
```

These commands invoke the type associated with `the(God?)` (namely that it satisfies the predicate `God?`), the premise `Greater1`, and then apply the standard automated proof strategy of PVS, called `grind`. Almost all the proofs mentioned subsequently are similarly straightforward and we do not reproduce them in detail.

As first noted by Garbacz [9], the premise `Greater1` begs the question under the other assumptions of the formalization, which we demonstrate by proving it from `God_re` and trichotomy of `>` (we need to duplicate and rename it as `Greater1_circ` because PVS does not allow forward references, as would be needed to cite `God_re` in a proof of `Greater1`). Given that we have proved `God_re` from `Greater1` and vice-versa, we can easily prove they are equivalent, as in `G1_begs_question`. Thus, in the definition of "begging the question" given earlier, C here is `God_re`, Q is `Greater1` and P is the rest of the formalization (i.e., `ExUnd`, the definition of `God?`, and the predicate subtype `trichotomous?` asserted for `>`).

[2] Trichotomy is the condition `FORALL x, y:  x > y OR y > x OR x = y`.

3 Begging the Question: Weaker Case

Eder and Ramharter [8, Section 2.3] claim that Anselm's Proslogion does not employ a definite description and that a correct reading is "*something* than which there is no greater." A suitable modification to the previous PVS theory is shown in Figure 3; the differences are that `>` is now an unconstrained relation on `beings`, and the conclusion is restated as the theorem `God_re_alt`. As before, this theorem is easily proved from the premises `ExUnd` and `Greater1` and the definition of `God?`. However, `Greater1` no longer strictly begs the question because it cannot be proved from the conclusion `God_re_alt`.

We can observe, however, that this specification of the Argument is very austere and imposes no constraints on the relation `>`; in particular, it could be an entirely empty relation. We demonstrate this in the theory interpretation `eandr1interp`, where all beings exist in reality, and none are `>` than any other (some may think this describes the real world), and beings are interpreted as natural numbers. PVS generates proof obligations (not shown here) to ensure the axioms of the theory `eandr1` are theorems under this interpretation: the interpretation of `Greater1` is trivially true, and we discharge the interpretation of `ExUnd` by exhibiting the number 42 as a witness.

Such a model seems contrary to the intent of the Argument: surely it is not intended that something than which there is no greater is so because nothing is greater than anything else. So we should surely require some minimal constraint on `>` to eliminate such vacuous models. A plausible constraint is that `>` be trichotomous; if we add this condition, as in `Greater1_circ_alt`, then the premise `Greater1` can again be proved from the conclusion `God_re_alt`. A weaker condition is to require only that beings satisfying the `God?` predicate are required to stand in the `>` relation to others; this is stated in `Greater1_circ_alt2` and is also sufficient to prove `Greater1` from `God_re_alt`.

In terms of the abstract formulation given at the beginning of Section 2, what we have here is that the conclusion C can be proved using the questionable premise Q: $P, Q \vdash C$, but not *vice versa*. However, if we augment the other premises P by adding some P_2, then we can indeed prove Q: $P, P_2, C \vdash Q$, and also the equivalence of C and Q: $P, P_2 \vdash C = Q$. Thus, Q does not beg the question C under the original premises P but does do so under the augmented premises P, P_2. We will say that Q *weakly begs* the question, where P_2 determines the "degree" of weakness.

Observe that one can always construct a P_2; the question is whether it is plausible and innocuous in the intended interpretation. Note, too, that discovery of suitable P_2 has some resemblance to use of abduction (to generate explanations or diagnoses) in AI, and that the discovery and formulation of existence axioms to prove theorems of mathematical systems is a program known as "reverse mathematics" [18].

```
eandr1: THEORY
BEGIN

  beings: TYPE
  x, y: VAR beings

  >(x, y): bool

  God?(x): bool = NOT EXISTS y: y > x

  re?(x): bool

  ExUnd: AXIOM EXISTS x: God?(x)

  Greater1: AXIOM FORALL x: (NOT re?(x) => EXISTS y: y > x)

  God_re_alt: THEOREM EXISTS x: God?(x) and re?(x)

%---------------- Question Begging Analysis ----------------------

  Greater1_circ_alt: THEOREM trichotomous?(>) =>
    (FORALL x: (NOT re?(x) => EXISTS y: y > x))

  Greater1_circ_alt2: THEOREM (FORALL x,y: God?(x) => x>y or x=y) =>
    (FORALL x: (NOT re?(x) => EXISTS y: y > x))

END eandr1

eandr1interp: THEORY
BEGIN

IMPORTING eandr1{{
  beings := nat,
  > := LAMBDA (x, y: nat): FALSE,
  re? := LAMBDA (x: nat): TRUE
}}
AS model

END eandr1interp
```

Figure 3: Eder and Ramharter's First Order Treatment, in PVS

4 Indirectly Begging the Question

Eder and Ramharter consider `Greater1` an unsatisfactory premise because it does not express "conceptions presupposed by the author" (i.e., Anselm) [8, Section 3.2] and says nothing about what it means to be *greater* other than the contrived connection to *exists in reality*. They propose an alternative premise `Greater2`, which is shown in Figure 4. This theory is the same as that of Figure 3, except that `Greater2` is substituted for `Greater1`, and a new premise `Ex_re` is added.

```
eandr2: THEORY
BEGIN

  beings: TYPE
  x, y: VAR beings

  >(x, y): bool

  God?(x): bool = NOT EXISTS y: y > x

  re?(x): bool

  ExUnd: AXIOM EXISTS x: God?(x)

  Ex_re: AXIOM EXISTS x: re?(x)

  Greater2: AXIOM FORALL x, y: (re?(x) AND NOT re?(y) => x > y)

  God_re_alt: THEOREM EXISTS x: God?(x) and re?(x)

END eandr2
```

Figure 4: Eder and Ramharter's Adjusted First Order Treatment, in PVS

It is easy to prove the conclusion `God_re_alt` from the new premises; they also directly entail `Greater1` so there is circumstantial evidence that they are question begging. However, it is not possible to prove `Greater2` from `God_re_alt` and the other premises, nor have I found a plausible augmentation to the premises that enables this. Thus, it seems that `Greater2` does not beg the question under our current definitions, neither strictly nor weakly.

But notice that we can always ensure that a premise does not beg the question by making it more general than necessary: that is, if Q begs the question, we replace it by some Q_2 such that $Q_2 \supset Q$; then $P, Q_2 \vdash C$ but for suitably chosen Q_2, $P, C \not\vdash Q_2$. It may be feasible to construct an augmentation P_2 to the premises so that $P, P_2, C \vdash Q_2$, so Q_2 is exposed as weakly begging, but P_2 might not be an attractively plausible premise (of course, Q_2 might not be either) so we should look for an alternative way to indict Q_2.

When constructing a mechanically checked proof of `God_re_alt` using `Greater2` I was struck how neatly the premise exactly fits the requirement of the proof at its penultimate step. To see this, observe the PVS sequent shown below; we arrive at this point following a few straightforward steps in the proof of `God_re_alt`. First, we introduce the premises `ExUnd` and `Ex_re`, expand the definition of `God?`, and perform a couple of routine steps of Skolemization, instantiation, and propositional simplification.

PVS Sequent

```
God_re_alt :

[-1]  re?(x!1)
  |-------
{1}   x!1 > x!2
[2]   re?(x!2)
```

PVS represents its current proof state as the leaves of a tree of sequents (here there is just one leaf); each sequent has a collection of numbered formulas above and below the `|-----` turnstile line; the interpretation is that the conjunction of formulas above the line should entail the disjunction of those below. Bracketed numbers on the left are used to identify the lines, and braces (as opposed to brackets) indicate this line is new or changed since the previous proof step. Terms such as `x!1` are Skolem constants. PVS eliminates top level negations by moving their formulas to the other side of the turnstile. Thus the sequent above is equivalent to the following.

Variant Sequent

```
God_re_alt :

[-1]  re?(x!1)
[2]   NOT re?(x!2)
  |-------
{1}   x!1 > x!2
```

We can read this as

```
re?(x!1) AND NOT re?(x!2) IMPLIES x!1 > x!2
```

and then observe that `Greater2` is its universal generalization.

PVS has capabilities that help mechanize this calculation. If we ask PVS to generalize the Skolem constants in the original sequent, it gives us the formula

```
FORALL (x_1, x_2: beings): re?(x_2) IMPLIES x_2 > x_1 OR re?(x_1)
```

Renaming the variables and rearranging, this is

```
FORALL (x, y: beings): (re?(x) AND NOT re?(y)) IMPLIES x > y
```

which is identical to `Greater2`. Thus, `Greater2` corresponds *precisely* to the formula required to discharge the final step of the proof.

I will say that a premise *indirectly* begs the question if it supplies exactly what is required to discharge a key step in the proof. Unless they are redundant or superfluous, all the premises to a proof will be essential to its success, so it may seem that any premise can be considered to indirectly beg the question. Furthermore, if we do enough deduction, we can often arrange things so that the final premise to be installed exactly matches what is required to finish the proof. My intent is that the criterion for indirect begging applies only when the premise in question perfectly matches what is required to discharge a key (usually final) step of the proof when the preceding steps have been entirely routine. In this circumstance, I claim that the sequent constructed by the PVS prover is a good representation of our epistemic state after we have digested the other premises. If the questionable premise then supplies exactly what is required to complete the proof (by generalizing the sequent), then it appears reverse-engineered, and certainly eliminates any hope of surprise or insight. Hence, I consider it to beg the question.

Clearly, individual judgement should be involved here, but may be difficult to apply in more complex proofs where a premise is employed as part of a longer chain of deductions. In the following section I show how careful proof structuring can isolate the application of a premise and expose its question begging character.

5 Indirect Begging in More Complex Proofs

In search of a more faithful reconstruction of Anselm's Argument, Eder and Ramharter observe that Anselm attributes properties to beings and that some of these (notably *exists in reality*) contribute to evaluation of the *greater* relation [8, Section 3.3]. They formalize this by hypothesizing some class `P` of "*greater*-making" properties on beings and then define one being to be greater than another exactly when it has all the properties of the second, and more besides. This treatment is higher order because it involves quantification over properties, not merely individuals. This is seen in the definition of `>` in the PVS formalization of Eder and Ramharter's higher order treatment shown in Figure 5. Notice that `P` is a set (which is equivalent to a predicate in higher-order logic) of predicates on `beings`; in PVS a predicate in parentheses as in `F: VAR (P)` denotes the corresponding subtype, so that `F` is a variable ranging over the subsets of `P`. A more detailed description of this PVS formalization is provided elsewhere [14].

```
eandrho: THEORY
BEGIN

  beings: TYPE

  x, y, z: VAR beings

  re?: pred[beings]

  P: set[ pred[beings] ]

  F: VAR (P)

  >(x, y): bool = (FORALL F: F(y) => F(x)) & (EXISTS F: F(x) AND NOT F(y))

  God?(x): bool = NOT EXISTS y: y > x

  ExUnd: AXIOM EXISTS x: God?(x)

  Realization: AXIOM
    FORALL (FF:setof[(P)]): EXISTS x: FORALL F: F(x) = FF(F)

  God_re_ho: THEOREM member(re?, P) => EXISTS x: God?(x) AND re?(x)

END eandrho
```

Figure 5: Eder and Ramharter's Higher Order Treatment, in PVS

The strategy for proving `God_re_ho` is first to consider the being `x` introduced by `ExUnd`; if this being exists in reality, then we are done. If not, then we consider a new being that has exactly the same properties as `x`, plus existence in reality—this is attractively close to Anselm's own strategy, which is to suppose that very same being can be (re)considered as existing in reality. In the PVS proof this is accomplished by the proof step

PVS Proof Step
```
(name "X" "choose! z: FORALL F: F(z) = (F(x!1) OR F=re?)")
```

which names `X` to be such a being. Here, `x!1` is the Skolem constant corresponding to the `x` introduced by `ExUnd` and `choose!` is a "binder" derived from the PVS choice function `choose`, which is defined in the PVS Prelude. This `X` is some being that satisfies all the predicates of `x!1`, plus `re?`. Given this `X`, it is easy to complete the proof, except that PVS generates the subsidiary proof obligation shown below to ensure that the choice function is well-defined (i.e., there is such an `X`).[3]

PVS TCC
```
EXISTS (x: beings): (FORALL F: F(x) = (F(x!1) OR F = re?))
```

This proof obligation requires us to establish that there is a being that satisfies the expression in the `choose!`; it is generated from the predicate subtype specified for the argument to `choose` and is therefore called a PVS Typecheck Correctness Condition, or TCC.

Eder and Ramharter provide the axiom `Realization` for this purpose; it states that for any collection of properties, there is a being that exemplifies *exactly* those properties and, when its variable `FF` is instantiated with the term

{ `G: (P) | G(x!1) OR G=re?`},

it provides exactly the expression above. In other words, `Realization` is a generalization of the formula required to discharge a crucial step in the proof. Thus, I claim that the premise `Realization` indirectly begs the question in this proof.

An alternative and more common style of proof in PVS would invoke the premise `Realization` directly at the point where `name` and `choose!` are used in the proof described here. The direct invocation obscures relationship between the formal proof and Anselm's own strategy, and it also uses `Realization` as one step in a chain of deductions that masks its question begging character. Thus, use of `name` and `choose!` are key to revealing both the strategy of the proof and the question begging character of `Realization`.

Campbell [7], who is updating his earlier book on the Argument [6], adopts some of Eder and Ramharter's higher order treatment, but rejects `Realization` on the grounds that it is false. Observe that we could have incompatible properties[4] and `Realization` would then provide the existence (in the understanding) of a being that exemplifies those incompatible properties, and this is certainly questionable. A better approach might be to weaken `Realization` to allow merely the addition of `re?` to the properties of some existing being. This is essentially the approach taken below.

---

[3] This is similar to the proof obligation generated for the definite description used in Oppenheimer and Zalta's rendition: there we had to prove that the predicate in `the` is uniquely satisfiable; here we need merely to prove that the predicate in `choose!` is satisfiable. The properties of the definite description, the choice function, and Hilbert's ε are described and compared in our description of Oppenheimer and Zalta's treatment [13].

[4] Eder and Ramharter are careful to require that all the greater-making properties are "positive" so directly contradictory properties are excluded, but we can have positive properties that are mutually incompatible [10]. Examples are being "perfectly just" and "perfectly merciful": the first entails delivering exactly the "right amount" of punishment, while the latter may deliver less than is deserved.

```
campbell: THEORY
BEGIN

  beings: TYPE

  x, y, z: VAR beings

  re?: pred[beings]

  P: set[ pred[beings] ]

  F: var (P)

  >(x, y): bool = (FORALL F: F(y) => F(x)) & (EXISTS F: F(x) AND NOT F(y))

  God?(x): bool = NOT EXISTS y: y > x

  ExUnd: AXIOM EXISTS x: God?(x)

  quasi_id(D: setof[(P)])(x,y: beings): bool =
     FORALL (F:(P)): NOT D(F) => F(x) = F(y)

  jre: setof[(P)] = singleton(re?)

  Weak_real: AXIOM
      NOT re?(x) => (EXISTS z: quasi_id(jre)(z, x) AND re?(z))

  God_re_ho: THEOREM member(re?, P) => EXISTS x: God?(x) AND re?(x)

END campbell
```

Figure 6: Simplified Version of Campbell's Treatment, in PVS

Campbell's formal treatment [7] differs from others considered here in that he includes more of Anselm's presentation of the Argument (e.g., where he speaks of "the Fool"). The treatment shown in Figure 6 is my simplified interpretation of Campbell's approach, scaled back to resemble the other treatments considered here. Campbell adopts Eder and Ramharter's higher order treatment, but replaces `Realization` by (in my interpretation) the axiom `Weak_real` which essentially states that if `x` does not exist in reality, then we can consider a being just like it that does. A being "just like it" is defined in terms of a predicate `quasi_id` introduced by Eder and Ramharter [8, Section 3.3] and is true of two beings if they have the same properties, except possibly those in a given set `D`. Observe that the PVS specification writes this higher order predicate in Curried form. Here, `D` is always instantiated by the singleton set `jre` containing just `re?`, so we always use `quasi_id(jre)`.

A couple of routine proof steps bring us to the following sequent.

```
God_re_ho :

{-1}  P(re?)
  |-------
[1]   EXISTS y: y > x!1
[2]   re?(x!1)
```

Our technique for discharging this is to instantiate formula 1 with a being just like `x!1` that does exist in reality, which we name `X`.

```
 (name "X" "(choose! z: quasi_id(jre)(z, x!1) AND re?(z))")
```

The main branch of the proof then easily completes and we are left with the obligation to ensure that application of the choice function is well-defined. That is, we need to show

```
EXISTS (z: beings): quasi_id(jre)(z, x!1) AND re?(z)
```

under the condition `NOT re?(x!1)`. This is precisely what the premise `Weak_real` supplies, so we may conclude that this premise indirectly begs the question.

The higher order formalizations considered in this section have slightly longer and more complex proofs than those considered earlier. This means that the indirect question begging character of a particular premise may not be obvious if it occurs in the middle of a chain of proof steps. Use of the `name` and `choose!` constructs accomplishes two things: it highlights the strategy of the proof (namely, it identifies the attributes of the alternative being to consider if the first one does not exist in reality), and it isolates application of the questionable premise to a context where its indirect question begging character is obvious.

6 Discussion

Once we go beyond the strict case, where the conclusion is directly equivalent to one of its premises, the idea of begging the question is open to discussion and personal judgement. A variety of positions are contested in the literature on argumentation (e.g., [2, 3, 17, 20–22]); however, I have not seen any discussion of question begging in fully formal settings. My proposal is that a premise may be considered to beg the question when it is equivalent to the conclusion, given the other premises (strict begging), or a light augmentation of these (weak begging), or when it directly discharges the key step of the proof (indirect begging). The intuition is that such premises are so close to the conclusion or its proof that they eliminate any sense of surprise. Strict begging is unequivocal; weak begging is rather less so: human judgement must determine whether the augmentation required to reveal question begging is innocuous or contrived. Indirect begging is still more equivocal: human judgement must determine whether the deductive effort prior to revelation of the perfect fit between premise and key proof step is routine or contrived.

It is not easy to compare these formal interpretations of begging the question to informal interpretations. Walton [20] discusses "epistemic" and "dialectical" models of question begging and uses the Araucaria diagramming system and other tools to present the structure of informal arguments. However, these are still far from fully formal, never mind mechanically analyzed, presentations. The essence of the epistemic model of begging the question is that truth of the premise in question is difficult to know or believe independently of the conclusion. Certainly, this must apply to our formulation of strict question begging, since the premise(s) concerned are directly equivalent to the conclusion. I think it also can apply to our formulation of weak question begging: if the augmentation to the other premises necessary to reveal weak begging is also required to establish a context in which the questionable premise(s) are credible (as in our example of Figure 3), then the questionable premise(s) would seem to beg the question in the epistemic sense.

Indirect begging is revealed when the questionable premise(s) supplies (a generalization of) exactly what is required to complete (or make a key move in) a proof. I claim that the proof state at that point (conveniently represented as a sequent) is a good summary of our current state of knowledge (i.e., our epistemic state); an indirectly begging premise is equivalent to (or a generalization of) one that can be reverse engineered from that state and so it seems that belief in such a premise cannot be independent of belief in the current proof state, which itself is a routine deduction from the conclusion and other premises.

Most authors who examine question begging in the Ontological Argument implicitly apply an epistemic criterion, and do so in the context of modal representations of the argument. Walton, however, does briefly consider first-order formulations in a paper that is otherwise about modal formulations [19].

Walton begins with a first-order formulation that is identical (modulo notation) to that of Figure 4. He asserts that the premise `Greater2` (his premise 2) is implausibly strong because it "would appear to imply, for example, that a speck of dust is greater than Paul Bunyan[5]." Walton suggests that premise `Greater1` of Figure 3 (his premise 2G) may be preferable but worries that our reason for believing `Greater1` must be something like `Greater2`. It is interesting that Walton does not indicate concern that `Greater1` might come close to begging the question, whereas our analysis shows that it is weakly begging (and becomes strictly begging

[5]Paul Bunyan is a lumberjack character in American folklore.

in the presence of premises that require a modicum of connectivity in the > relation). Thus, I suggest that the formulations and methods of analysis proposed here are more precise, informative, and checkable than Walton's informal interpretations for begging the question.

Begging the question is not a fatal defect; the equivalence, or near equivalence, to the conclusion of a strict or weakly question begging premise may be an interesting observation, but I do not think proof of the conclusion retains much interest once a premise has been shown to beg the question in these ways. Indirect begging is perhaps a somewhat different case: it reveals how exquisitely crafted—one is tempted to say reverse-engineered—is the premise to its rôle in the proof. To my mind, it casts doubt on the extent to which the premise may be considered analytic in the sense that Eder and Ramharter use the term: that is, something that the author "could have held to be true for conceptual (non-empirical) reasons" [8, Section 1.2(7)].

In the case of the Ontological Argument, I have shown that several first- and higher-order formalizations employ question begging premises. I suspect this is true of all similar formulations. Separately (in work performed after this paper was prepared) [15], I have examined several formulations of the argument in quantified modal logic (including that of Rowe [12], who explicitly accuses the Argument of begging the question, and that of Adams [1], who also discusses circularity) and found them vulnerable to the same criticism. The analysis there reveals that modal formulations of the Argument admit delicate choices in how the quantification is arranged and this is strongly related to identification of the premises accused (e.g., by Rowe) of question begging.

It is, of course, for individual readers to decide whether the forms of question begging that we identify affect their confidence, or their interest, in these various renditions of the Argument, and in the Argument itself.

Observe that detection of the various kinds of question begging requires exploration of variations on a specification or proof. This is tedious and error-prone to do by hand, but simple, fast, and reliable using mechanized assistance. I hope the methods and tools illustrated here will encourage others to investigate similar questions concerning this and other formalized arguments: as Leibniz said, "let us calculate."

References

[1] Robert Merrihew Adams. The logical structure of Anselm's arguments. *The Philosophical Review*, 80(1):28–54, 1971.

[2] John A. Barker. The fallacy of begging the question. *Dialogue*, 15(2):241–255, 1976.

[3] John A. Barker. The nature of question-begging arguments. *Dialogue*, 17(3):490–498, 1978.

[4] Christoph Benzmüller and Bruno Woltzenlogel Paleo. Gödel's God in Isabelle/HOL. *Archive of Formal Proofs*, 2013.

[5] Christoph Benzmüller and Bruno Woltzenlogel Paleo. Interacting with modal logics in the Coq proof assistant. In *Computer Science—Theory and Applications: 10th International Computer Science Symposium in Russia, CSR 2015*, Volume 9139 of Springer-Verlag *Lecture Notes in Computer Science*, pages 398–411, Listvyanka, Russia, July 2015.

[6] Richard J. Campbell. *From Belief to Understanding: A Study of Anselm's Proslogion Argument on the Existence of God.* Faculty of Arts, Australian National University, 1976.

[7] Richard J. Campbell. Personal communication, June 2016.

[8] Günther Eder and Esther Ramharter. Formal reconstructions of St. Anselm's ontological argument. *Synthese*, 192(9):2795–2825, October 2015.

[9] Paweł Garbacz. PROVER9's simplifications explained away. *Australasian Journal of Philosophy*, 90(3):585–592, 2012.

[10] Kenneth Einar Himma. Ontological argument. In James Fieser and Bradley Dowden, editors, *Internet Encyclopedia of Philosophy.* April 2005.

[11] Paul E. Oppenheimer and Edward N. Zalta. On the logic of the Ontological Argument. *Philosophical Perspectives*, 5:509–529, 1991. Reprinted in *The Philosopher's Annual: 1991*, Volume XIV (1993): 255–275.

[12] William L. Rowe. The Ontological Argument and question-begging. *International Journal for Philosophy of Religion*, 7(4):425–432, 1976.

[13] John Rushby. The Ontological Argument in PVS. In Nikolay Shilov, editor, *Fun With Formal Methods*, St Petersburg, Russia, July 2013. Workshop in association with CAV'13.

[14] John Rushby. Mechanized analysis of a formalization of Anselm's ontological argument by Eder and Ramharter. CSL technical note, Computer Science Laboratory, SRI International, Menlo Park, CA, January 2016.

[15] John Rushby. Mechanized analysis of modal reconstructions of Anselm's Ontological Arguments. Technical Report SRI-CSL-17-xx, Computer Science Laboratory, SRI International, Menlo Park, CA, 2017. In preparation.

[16] Bertrand Russell. *History of Western Philosophy: Collectors Edition*. Routledge, 2013.

[17] David H. Sanford. The fallacy of begging the question: A reply to Barker. *Dialogue*, 16(3):485–498, 1977.

[18] Stephen G. Simpson. Reverse mathematics. In *Proc. Symposia in Pure Mathematics*, Volume 42, pages 461–471, 1985.

[19] Douglas Walton. The circle in the Ontological Argument. *International Journal for Philosophy of Religion*, 9(4):193–218, 1978.

[20] Douglas Walton. Epistemic and dialectical models of begging the question. *Synthese*, 152(2):237–284, September 2006.

[21] Douglas N. Walton. *Begging the Question: Circular Reasoning as a Tactic of Argumentation*. Greenwood Press, New York, NY, 1991.

[22] Douglas N. Walton. Begging the question as a pragmatic fallacy. *Synthese*, 100(1):95–131, 1994.